In[10]:= `ff[x_] = Exp[-x^2]`

Out[10]= e^{-x^2}

In class we found that the parametric vector equation of the surface which determines a solution of Burgers' equation is

In[11]:= `{ff[ξ] s + ξ, s, ff[ξ]}`

Out[11]= $\{e^{-\xi^2} s + \xi, s, e^{-\xi^2}\}$

We can plot it using ParametricPlot3D

In[12]:= `ParametricPlot3D[{ff[ξ] s + ξ, s, ff[ξ]}, {s, 0, 3}, {ξ, -3, 3}, BoxRatios → {4, 1, 1}, AxesLabel → {x, t, z}]`

Out[12]=



And from this picture it is clear that this is not a graph of function of x and t. However, if we restrict time, we might get a graph of a function of x and t:

In[13]:= `ParametricPlot3D[{ff[ξ] s + ξ, s, ff[ξ]}, {s, 0, 1}, {ξ, -3, 3}, BoxRatios → {4, 1, 1}, AxesLabel → {x, t, z}]`

Out[13]=



This surface looks like a graph of a function. Next we will find the maximum value of t, that is s in our setting for which our surface will be a graph of a function. To this end we look at the cross sections of the surface with fixed time, that is fixed s. We study the parametric equation

In[14]:= `{ff[ξ] s + ξ, ff[ξ]}`

Out[14]= $\{e^{-\xi^2} s + \xi, e^{-\xi^2}\}$

in xz-plane. For this Manipulate[] is a great tool.

When you execute the Manipulate[] command, click on the + sign next to the "s" slider. That will give you the information about the s values that are used in the plot. We would like to find the smallest s for which the resulting graph is not a function, that is does not satisfy the vertical line test.

In[15]:=
```
Manipulate[ParametricPlot[{ff[ξ] s + ξ, ff[ξ]}, {ξ, -4, 4}, PlotRange → {{-4, 4}, {-0.1, 1.1}},
  GridLines → {Range[-4, 4, 0.2], Range[0, 1, 0.2]}, AspectRatio → 1 / 2, ImageSize → 500], {s, 0, 3}]
```

Out[15]=



Looking at the above manipulation, I conclude that at around $s \approx 1.205$ the graph is not a function. The vertical tangent line occurs around $x \approx 1.41$. These are my guesses that we will verify below.

The vertical tangent line will occur when the first component of the tangent vector is equal to 0. Let us calculate the tangent vector:

In[16]:=
```
D[{ff[ξ] s + ξ, ff[ξ]}, ξ]
```

Out[16]= $\left\{1 - 2\, e^{-\xi^2}\, s\, \xi,\ -2\, e^{-\xi^2}\, \xi\right\}$

For which s (time) the first component becomes 0?

In[17]:=
```
Solve[1 - 2 e^(-ξ^2) s ξ == 0, s]
```

Out[17]= $\left\{\left\{s \to \frac{e^{\xi^2}}{2\,\xi}\right\}\right\}$

Thus s must equal the above expression in ξ. Let us check that expression of ξ:

```
Plot[e^(ξ^2)/(2 ξ), {ξ, -3, 3}, PlotRange → {-4, 4},
  GridLines → {Range[-4, 4, 0.2], Range[-4, 4, 0.2]}]
```
We know that

Out[•]=



We want a positive s, the smallest positive s. In the above plot s is on the vertical axis. Let us find the smallest such s:

In[18]:= $D\left[\frac{e^{\xi^2}}{2\,\xi}, \xi\right]$

Out[18]= $e^{\xi^2} - \frac{e^{\xi^2}}{2\,\xi^2}$

In[19]:= $\text{Solve}\left[e^{\xi^2} - \frac{e^{\xi^2}}{2\,\xi^2} == 0, \xi\right]$

Out[19]= $\left\{\left\{\xi \to -\frac{1}{\sqrt{2}}\right\}, \left\{\xi \to \frac{1}{\sqrt{2}}\right\}\right\}$

Clearly we need $\xi = 1/\sqrt{2}$. Thus, the number below is the s that we seek

In[20]:= $\left(\frac{e^{\xi^2}}{2\,\xi}\right)$ /. $\left\{\xi \to \frac{1}{\sqrt{2}}\right\}$

Out[20]= $\sqrt{\frac{e}{2}}$

In[21]:= $N\left[\sqrt{\frac{e}{2}}\right]$

Out[21]= 1.16582

Ok, this is sufficiently close to what I guessed, around $s \approx 1.205$

Now check x. The x coordinate is in fact ff[ξ] s+ξ

In[22]:= (ff[ξ] s + ξ) /. $\left\{\xi \to \frac{1}{\sqrt{2}}, s \to \sqrt{\frac{e}{2}}\right\}$

Out[22]= $\sqrt{2}$

In[23]:= $N\left[\sqrt{2}\right]$

Out[23]= 1.41421

This is much closer to my guess, around $x \approx 1.41$.